
第 6 章  循环结构程序设计 

一、选择题 

6.1  阅读下面的程序段： 

For a=1 To 2 

  For b=1 To a 

    For c=b To 2 

      i=i+1 

    Next 

  Next 

Next 

Print i 

执行上面的三重循环后，i 的值为（   ）。 

A）4    B）5    C）6    D）9 

【解答】正确答案为 B。 

分析：本题是一个多重循环语句，是 For…Next 结构的三重嵌套，仔细分析一下就会看

到，在结构中 i=i+1 共执行了 5 次，所以最后 i 的值为 5。 

6.2  设有下面的循环： 

i=1 

Do 

  i=i+3 

  Print i 

Loop Until  i> ____ 

程序运行后要执行 3 次循环体，则条件中 i 的最小值为（   ）。 

A）6     B）7    C）8    D）9 

【解答】正确答案为 B。 

分析：本题考查的是 DO…LoopUntil 结构的循环语句，不管是否满足条件，都要先执行

一次。经分析当 i>7（或 8 或 9）都满足执行 3 次的条件，但题目要求最小的，所以应该是 7。 

6.3  下面程序段的运行结果是（   ）。 

a=1 

b=1 

Do 

  a=a+1 

  b=b+1 

Loop Until b>5 

Print"k=";a;Spc(4);"b=";b+a 

A）k=7 b=14   B）k=6 b=6   C）k=4 b=8   D）k=6 b=12 

【解答】正确答案为 D。 

分析：本题中 Do-Loop Until 循环为直到型循环结构，直到条件 b>5 为止，此时 a=6, b=6，

所以最后输出的结果是选项 D。 

6.4  下列程序段的执行结果为（   ）。 

a=6 

b=1 

For i=1 To 3 

  f=a+b 

  a=b 

  b=f 

  Print f; 

Next i 

A）2  3  6   B）2  3  5   C）2  3  4   D）2  2  8 

【解答】正确答案为 B。 

分析：本程序段的运行过程如下。 

开始时 a=1, b=1。For 循环中步长默认值为 1，循环变量 I 的初值为 1，终值为 3，所以

此循环结构可以循环 3 次。 

第一次循环后，结果为：f=2, a=1, b=2。 

第二次循环后，结果为：f=3, a=2, b=3。 

第三次循环后，结果为：f=5, a=3, b=5。 

每循环一次，输出 f 当前值，循环 3 次，即输出 3 个 f 值，分别为 2, 3, 5。 

6.5  在窗体上画一个名称为 Command1 的命令按钮和一个名称为 Text1 的文本框，然

后编写如下事件过程： 

Private Sub Command1_Click() 

  n=Val(Text1.Text) 

  For i=2 To n 

    For j=2 To Sqr(i) 

        If i Mod j=0 Then Exit For 

    Next j 

    If j>Sqr(i)Then Print i 

  Next i 

End Sub 

该事件过程的功能是（   ）。 

A）输出 n 以内的奇数         B）输出 n 以内的偶数 

C）输出 n 以内的素数       D）输出 n 以内能被 j 整除的数 

【解答】正确答案为 C。 

分析：要理解本事件过程的作用就要看最后打印的条件：j>Sqr(i)，而最后这个条件是否

成立又取决于内循环，内循环的作用就是判断在 2 到某个数的平方根之间是否有它本身的因

数，显然这是判断素数的条件。 

6.6  下列程序段的执行结果为（   ）。 

a=5 

For k=1 To 0 

  a=a + k 

Next k 

Print k;a 

A）1  6   B）1  16   C）1  5    D）11  21 

【解答】正确答案为 C。 

分析：For 循环语句的执行过程为：“循环变量”首先取得“初值”，检查是否超过“终

值”，如果超过，就一次也不循环而跳出循环，属于“先检查后执行”的类型。在本程序段

的 For k=1 To 0 语句中，初值为 1，终值为 0，显然当“循环变量”首先取得“初值”1，检

查后超过“终值”0，所以一次也不执行，即最后执行 Print 时，k=1, a=5。 

6.7  下列程序段的执行结果为（   ）。 

a=3 

b=1 

For i=1 To 3 

  f=a+b 

  a=b 

  b=f 

  Print f ;  

Next i 

A）4 3 6    B）4 5 9    C）6 3 4    D）7 2 8 

【解答】正确答案为 B。 

分析：本程序段的运行过程如下。 

开始时 a=3, b=1。For 循环中步长默认值为 1，循环变量 i 的初值为 1，终值为 3，所以

此循环结构可以循环 3 次。 

第一次循环后，结果为：f=4, a=1, b=4。 

第二次循环后，结果为：f=5, a=4, b=5。 

第三次循环后，结果为：f=9, a=5, b=9。 

所以每循环一次，便输出 f 当前值，循环 3 次，即输出 3 个 f 值，分别为 4, 5, 9。 

6.8  下列程序段的执行结果为（   ）。 

i=9 

x=5 

Do 

  i=i+1 

  x=x+2 

Loop Until i >=7 

Print "i="; i; 

Print "x="; x 

A）i=4 x=5   B）i=7 x=15   C）i=6 x=8   D）i=10 x=7 

【解答】正确答案为 D。 

分析：本题用的是 Do 循环结构，此循环由于“先执行后检查”，所以至少执行一次。本

题中，程序运行到循环条件 i>=7 的值为 True 时，才停止。所以当程序结束运行后 i=10，x=7。 

6.9  执行下列程序段后，输出的结果是（   ）。 

For k1=0 To 4 

  y=20 

  For k2=0 To 3 

    y=10 

    For k3=0 To 2 

      y=y + 10 

    Next k3 

  Next k2 

Next k1 

Print y 

A）90    B）60    C）40    D）10 

【解答】正确答案为 C。 

分析：这是一个三重循环嵌套的程序。多重 For 循环的执行过程是：外层循环的循环控

制变量每取一个值，内循环的循环控制变量要取遍所有的值。这个程序最外两层循环的最后

一次循环执行时 k1=4, k2=3，都要再执行一次 y=10，不管 y 以前为何值，在这里 y 都要重新

取得初值 10，这时再执行一次内循环。内层循环体要执行 3 次，即加 3 次 10。y 最终的值

是 40。 

6.10  新建一个列表框，要实现对列表项可以复选，应设置的属性是（   ）。 

A）ScrollBars   B）MultiSelect  C）DataField   D）Stretch 

【解答】正确答案为 B。 

分析：在列表框的属性中，可以通过设置 MultiSelect 属性来实现是否可以进行列表项的

多选，属性值为 0，不可以多选列表；属性为 1，不需要用 Ctrl 键或者 Shift 键就可进行多选；

属性值为 2，要利用 Ctrl 键或者 Shift 键，所以选项 B 是正确的。而选项 A 是设置滚动条的，

与此无关。选项 C 中 DataField 是用来设置数据的，也无关。选项 D 中 Stretch 不是列表框的

属性，所以也不正确。 

6.11  在窗体上画一个命令按钮，然后编写如下事件过程： 

Private Sub Command1_Click() 

  x=0 

  Do Until x=l 

    a=InputBox("请输入 A 的值") 

    a=Val(A) 

    b=InputBox("请输入 B 的值") 

    b=Val(B) 

    x=InputBox("请输入 X 的值") 

    x=Val(X) 

    a=a+bx 

  Loop 

  Print a 

End Sub 

程序运行后，单击命令按钮，依次在输入对话框中输入 5、4、3、2、l、1，则输出结

果为（   ）。 

【解答】正确答案为 2。 

分析：本题需要注意的是，每次循环开始时 3 个变量都被重新赋值一次，所以最后的结

果仅仅是 2+1+(1)。所以最终的输出是 2。 

6.12  在 Visual Basic 中，组合框是文本框和（    ）特性的组合。 

A）复选框   B）标签    C）列表框   D）目录列表框 

【解答】正确答案为 C。 

分析：组合框是一个独立的控件，它具有列表框和文本框的功能，它可以像列表框一样，

让用户通过鼠标选择需要的项目，也可以像文本框一样，用输入的方式选择项目。 

6.13  下列语句中，返回列表框 List1 中项目个数的语句是（   ）。 

A）x=List1.ListCount      B）x=ListCount 

C）x=List1.ListIndex      D）x=ListIndex 

【解答】正确答案为 A。 

分析：List 控件的 ListCount 属性返回列表部分项目的个数。 

二、填空题 

6.14  执行下面的程序段，x 的值为____。 

Private Sub Command1_Click() 

  For i=1 To 9 

    a=a + i 

  Next i 

  x=Val(i) 

  MsgBox x 

End Sub 

【解答】正确答案为 10。 

分析：在程序中 For…Next 语句之后并没有对 i 的值改变，此时 i 为 10，故运行后结果

为 10。 

6.15  以下程序的功能是从键盘输入若干个学生的考试成绩，统计并输出最高分和最低

分，当输入负数时结束输入，输出结果。请补全程序。 

Dim x,amax,amin As Single 

x=InputBox("Enter a score") 

amax=x 

amin=x 

Do While  ____ 

  If x>amax Then 

    amax=x 

  End If 

  If  _____  Then 

    amin=x 

  End If 

  x＝InputBox("enter a score") 

Loop 

Print "max="; amax, "min="; amin 

【解答】正确答案为：x>=0；x<amin。 

分析：本程序先定义了 3 个变量 x、amin、amax，它们分别用来接收从键盘输入的数字、

最小值和最大值，先用 InputBox 函数从键盘接收一个整数并赋给 x。然后将 x 的值赋给 amin

和 amax，作为它们的初值。因为当输入的值为负数时结束循环，所以 Do While 的控制语句

就是判断 x 的值是否大于等于“0”。当 x≥0 时，执行块结构条件语句；当 x>amax 时，将 x

的值赋给 amax；显然，当 x<amin 时，就将 x 的值赋给 amin。每循环一次，就要给 x 赋值一

次，然后接着循环直到 x 的值为负数时就跳出循环，输出 amax 和 amin 的值。 

6.16  下面程序用来打印乘法“九九表”，请补全程序。 

Dim i As Integer, j As Integer, Str1$ 

Str ="" 

For i=1 To 9 

  For j=1 To 9  

      If  ____  Then 

        Str1=Str1+Str$（j）+""+Str$(i)+ "="+Str$(Val(i * j)) 

      Else 

        Str1=Str1 & Chr(13) 

         _____ 

      End If 

  Next j 

Next i 

Print Str1 

【解答】正确答案为 j < = i ；Exit FOR。 

分析：Exit 语句总是出现在 If 语句或 Select Case 语句内部，而 If 语句或 Select Case 语

句在循环内嵌套，用 Exit 语句中断循环。函数首先定义了 Integer 型变量 i、j，并将空格赋给

Str1。第一个 For 循环的变量 i 从 1 到 9 步长为 1，第二个循环的变量 i 也是从 1 到 9 步长为

1。循环体为选择结构，它用来输出 i*j 的值，所以 j 的值应该小于此时 i 的值，那么 If 的判

断语句为 j < = i。当满足条件时，执行 Then 后面的语句，即输出 i*j 的值；如果不满足，则

执行 Else 后面的语句，并跳出内循环。 

6.17  在窗体上画一个命令按钮，然后编写如下程序： 

Function fun(By Val num As Long)As Long 

  Dim k As Long 

  k=l 

  num=Abs(num) 

  Do While num 

    k=k*(num Mod 10) 

    num=num \ 10 

  Loop 

  fun=k 

End Function 

Private Sub Command1_Click() 

  Dim n As Long , r As Long 

  n=InputBox("请输入一个数") 

  n=CLng(n)  :  r=fun(n) 

  Print r 

End Sub 

程序运行后，单击命令按钮，在输入对话框中输入"345"，输出结果为____。 

【解答】正确答案为 60。 

分析：程序首先要求用户输入一个数，然后将这个数转化为长整型数传递给函数 fun。

本题中传递的数字是 345。函数 fun 首先对传来的参数取绝对值，然后进入循环。 

第一次循环结束的时候：k=5，num=34。 

第二次循环结束的时候：k=20，hum=3。 

第三次循环结束的时候：k=60，num=0。 

6.18  在窗体上画一个命令按钮，然后编写如下事件过程： 

Private Sub Command1_Click() 

  For i=l To 4 

    x=4 

    For j=1 To 3 

      x=3 

      For k=1 To 2 

        x=x+6 

      Next k 

    Next j 

  Next i 

  Prim x 

End Sub 

程序运行后，单击命令按钮，输出结果是____。 

【解答】正确答案为 15。 

分析：观察程序，由于每次执行第二层循环时 x 都被重新赋值，因此只需分析 x=3 时，

执行第三层循环所得到的结果就可以了。 

k=l，x=x+6=3+6=9 

k=2，x=x+6=9+6=15 

最终的输出结果是：15。 

6.19  以下程序的功能是：从键盘上输入若干个数字，当输入负数时结束输入，统计出

全部数字的平均值，输出结果。请补全程序。 

Private Sub Form_Click() 

  Dim x , y As Single , z As Integer 

  x=InputBox("Enter a score") 

  Do while  ____ 

    y=y+x 

    z=z+1 

    x=InputBox("Enter a score") 

  Loop 

  If z=0 Then 

    z=l 

  End If 

  y=  ____ 

  Print y 

End Sub 

【解答】正确答案为：x>=0；y/z。 

解析：本题的循环首先判断输入是否大于 0，如果是，则处理这个输入，否则结束循环。

循环体内的第二个判断语句判断当前输入是否比当前的最小值小，如果是，则改写当前最小

值为输入值，否则结束判断语句。 

三、编程题 

6.20  输入初始值，输出 100 个不能被 3 整除的数。 

【解答】 

1）设计程序界面和设置对象属性，如图 6-1 所示。 

2）编写代码。 

增加窗体的 Activate 事件代码： 
 

图 6-1  输出 100 个不能被 3 整除的数 

Private Sub Form_Activate( ) 

  Text2.SetFocus 

End Sub 

改写命令按钮 Command1 的 Click 事件代码： 

Private Sub Command1_Click( ) 

  Dim x As Integer, n As Integer 

  x = Val(Text2.Text): n = 1 

  Do Until n > 100 

    If x Mod 3 < > 0 Then 

      Text1.Text = Text1.Text & Str(x) & Chr(13) & Chr(10) 

      n = n + 1 

    End If 

    x = x + 1 

  Loop 

End Sub 

6.21  设计程序，求 s = 1 + (1 + 2) + (1 + 2 + 3) + … + (1 + 2 + 3 + … + n)的值。 

【解答】本题先进行内循环累加，再进行外循环累加，因此，需要两个累加器。但是，

由于本题较特殊，也可以用一个循环来实现。括号内每累加一个数，就往 s 中累加一次，其

结果一样。 

1）设计程序界面并设置对象属性，如图 6-2（a）所示。其中 Command1 的 Default 属性

为 True（窗体的默认命令按钮，使得按 Enter 键等于单击命令按钮）。 

      
（a）                                    （b） 

图 6-2  程序界面与程序运行结果 

2）编写代码。 

首先编写窗体的 Activate 事件代码： 

Private Sub Form_Activate( ) 

  Text1.SetFocus 

End Sub 

编写命令按钮 Command1 的 Click 事件代码： 

Private Sub Command1_Click( ) 

  Dim n As Integer, i As Integer, j As Long, s As Long 

  n = Val(Text1.Text) 

  j = 0:  s = 0 

  For i = 1 To n 

    j = j + i 

    s = s + j 

  Next i 

  Label2.Caption = "和数 s = " & Str(s) 

  Text1.SelStart = 0 

  Text1.SelLength = Len(Text1.Text) 

End Sub 

运行程序，输入一个整数后按 Enter 键，显示结果如图 6-2（b）所示。 

6.22  设 s = 112233…nn，求 s 不大于 400 000 时最大的 n 值。 

【解答】本题先进行内循环累乘（求幂），再进行外循环累乘，因此需要两个累乘器。计

数器为 n，外累乘器 s = s * t，内累乘器 t = t * n。其循环条件是 s <= 400000，又由于求的是

最大的 n 值，因此输出语句应在外循环体外。 

窗体界面的设计与程序运行的结果，如图 6-3 所示。 

      
（a）                                    （b） 

图 6-3  程序界面与程序运行结果 

编写命令按钮的 Click 事件代码： 

Private Sub Command1_Click( ) 

  Dim n As Integer, s As Long 

  CurrentY = Label1.Height + 400 

  Print " n", " s" 

  Print 

  n = 1 

  s = 1 

  Do While s <= 400000 

    n = n + 1 

    t = 1 

    For i = 1 To n 

      t = t * n 

    Next 

    s = s * t 

    Print n, s                        ' 通过本行可以看到循环过程 

  Loop 

  Label2.Caption = "n = " & Str(n – 1) 

End Sub 

6.23  设有一张厚为 x 毫米、面积足够大的纸，将它不断地对折。试问对折多少次后，

其厚度可达珠穆朗玛峰的高度（8848 米）。 

【解答】设对折后纸的厚度为 h 毫米，计数器为 n。在没有对折时，纸厚为 x 毫米，每对

折一次，其厚度是上一次的 2 倍，在未到达 8848 米时，重复进行对折。还要注意，在解题

时要把单位换算一致。 

1）建立应用程序用户界面并设置对象属性，如图 6-4（a）所示。 

       
（a）                                       （b） 

图 6-4  用户界面与程序运行结果 

2）编写程序代码。根据流程图，写出命令按钮 Command1 的 Click 事件代码为： 

Private Sub Command1_Click( ) 

  n = 0                               ' 对折次数 

  h = Text1.Text                        ' 纸的厚度 

  Do While h < 8848000 

    n = n + 1                         ' 对折次数 

    h = 2 * h                         ' 对折 

  Loop 

  Text2.Text = n 

End Sub 

如果改为直到型循环结构，程序代码为： 

Private Sub Command1_Click( ) 

  n = 0                               ' 对折次数 

  h = Text1.Text                       ' 纸的厚度 

  Do 

    n = n + 1                          ' 对折次数 

    h = 2 * h                          ' 对折 

  Loop Until h >= 8848000 

  Text2.Text = n 

End Sub 

运行结果如图 6-4（b）所示。 

6.24  我国古代数学家张丘建在“算经”里曾提出一个世界数学史上有名的百鸡问题：

“鸡翁一，值钱五；鸡母一，值钱三；鸡雏三，值钱一；百钱买百鸡，问鸡翁、母、雏各几

何？”请编写程序，求出结果。 

【解答】设公鸡 x 只，母鸡 y 只，小鸡 z 只，依题意可以列出以下方程组： 
100

5 3 100
3

x y z
zx y

　  



  

 

在这两个方程中，由于有 3 个未知数，属于不定方程，无法直接求解。下面用“穷举法”，

将各种可能的组合全部一一测试，将符合条件的组合输出。 

先设 x = 1，y = 1，则 z = 100 – x – y = 98，检查这一组的价钱加起来是否为 100 元。经验

算，不等于 100 元，所以这一组不符合要求。再看下一组，仍保持 x = 1，而 y = 2，则 z = 100 – 

1 – 2 = 97，价钱为 51 + 32 + 97/3≈43，也不符合要求。保持 x = 1，使 y 由 3 变到 100，依

次测试各组合。 

然后设 x = 2，y 再由 1 变到 100 …… 直到 x = 100，y 再由 1 变到 100。这样就把全部可

能的组合一一测试过了。 

（1）建立应用程序用户界面并设置对象属性，如图 6-5 所示。 

（2）编写程序代码。命令按钮 Command1 的 Click 事件代码为： 

Private Sub Command1_Click( ) 

  List1.Clear 

  For x = 1 To 100 

    For y = 1 To 100 

      z = 100 – x – y 

      If 5 * x + 3 * y + z / 3 = 100 Then 

        p = Format(x, "@@@@") & Format(y, "@@@@@") & Format(z, "@@@@@") 

        List1.AddItem p 

      End If 

    Next 

  Next 

End Sub 

经过对上述代码的分析可知，实际上并不需要使 x 从 1 变到 100，y 从 1 变到 100。因为

公鸡每只 5 元，100 元最多买 20 只公鸡，而如果 100 元全买了 20 只公鸡的话，就买不了母

鸡和小鸡了，不符合“百钱买百鸡”的要求。所以公鸡不可能是 20 只，最多只能买 19 只。

同理，母鸡一只 3 元，100 元最多买 33 只。因此，可将代码修改为： 

Private Sub Command1_Click( ) 

  List1.Clear 

  For x = 1 To 19 

    For y = 1 To 33 

      z = 100 – x – y 

      If 5 * x + 3 * y + z / 3 = 100 Then 

 
图 6-5  百钱买百鸡 

        p = Format(x, "@@@@") & Format(y, "@@@@@") & Format(z, "@@@@@") 

        List1.AddItem p 

      End If 

    Next 

  Next 

End Sub 

6.25  打印乘法“九九表”，输出结果分别如图 6-6（a）、（b）所示。 

     
（a）                                               （b） 

图 6-6 “九九表” 

【解答】若要输出图 6-6（a）所示的“九九表”，其窗体的 Load 事件代码为： 

Private Sub Form_Load( ) 

  Show 

  FontSize = 12  

  Print Tab(25); "九九表"  

  Print 

  Print " * "; 

  For i = 1 To 9                 ' 输出第一行数字（1～9） 

    Print Tab(i * 6); i; 

  Next i 

  Print   

  For j = 1 To 9                 ' 外层循环 

    Print j; " "; 

    For k = 1 To j               ' 注意，内层循环的初值为 1，终值为 j 

      m = j * k  

      Print Tab(k * 6); m; " ";   

    Next k 

    Print   

  Next j 

End Sub 

若要输出图 6-6（b）所示的“九九表”，其窗体的 Load 事件代码为： 

Private Sub Form_Load( ) 

  Show 

  FontSize = 12   

  Print Tab(25); "九九表" 

  Print  

  Print " * "; 

  For i = 1 To 9                 ' 输出第一行数字（1～9） 

    Print Tab(i * 6); i; 

  Next i 

  Print 

  For j = 1 To 9                 ' 外层循环 

    Print j; " "; 

    For k = j To 9               ' 注意，内层循环的初值为 j，终值为 9 

      m = j * k  

      Print Tab(k * 6); m; " ";   

    Next k 

    Print 

  Next j 

End Sub 

6.26  所谓“水仙花数”，是指一个 3 位数，其各位数的立方和等于该数，如 153 = 13 + 
53 + 33，编写程序输出所有的“水仙花数”。 

【分析】此题的关键是把任意 3 位数的每一位数分离出来。设 a, b, c 分别是 3 位整数 n

的百位数、十位数、个位数，则 

a = Int(n / 100) 

b = Int((n – a * 100) / 10) 

c = n – a * 100 – b * 10 

【解答】窗体界面的设计如图 6-7 所示。 

命令按钮的 Click 事件代码为： 

Private Sub Command1_Click( ) 

  Dim p As Integer 

  List1.Clear 

  For n = 100 To 999 

      a = Int(n / 100) 

      b = Int((n - a * 100) / 10) 

      c = n - (a * 100 + b * 10) 

      p = a ^ 3 + b ^ 3 + c ^ 3 

      If p = n Then List1.AddItem p 

  Next 

End Sub 

6.27  求 1000～1100 之间的所有素数。 

【分析】对 1000～1100 之间的各整数依次测试其是否为素数，需使用双重循环。 

1）建立用户界面并设置对象属性，如图 6-8 所示。 

 
图 6-7  求水仙花数 

2）编写事件代码。 

编写“开始”命令按钮 Command1 的 Click 事件代码： 

Private Sub Command1_Click( ) 

  For n = 1001 To 1100 Step 2 

    s = 0 

    For i = 2 To Int(Sqr(n)) 

      If n Mod i = 0 Then 

        s = 1 

        Exit For 

      End If 

    Next 

    If s = 0 Then List1.AddItem n 

  Next 

End Sub 

6.28  利用下述公式计算圆周率的近似值： 
1 1 1 1

1
4 3 5 7 9

        

当最后一项的绝对值小于 0.000 001 时停止计算。 

【解答】在立即窗口中输出计算结果的事件代码如下： 

Private Sub Command1_Click( ) 

  Dim m!, s#, t#, k! 

  k = 1: s = 1: t = 1: m = 1 

  While Abs(t) > 0.0000001 

    k = k + 2 

    m = -m 

    t = m / k 

    s = s + t 

  Wend 

  Debug.Print Format(4 * s, "0.0000000") 

End Sub 

6.29  在窗体上输出如图 6-9 所示的图形。 

【解答】 

1）建立用户界面并设置对象属性，如图 6-9 所示。 

2）编写命令按钮的 Click 事件代码： 

Private Sub Command1_Click( ) 

  Cls 

  Print 

  For n = 1 To 5  

    Print Tab(2 * n + 2); 

    For m = n To 10 – n 

 
图 6-8  求素数 

 
图 6-9  输出图形 

      Print Spc(1); "*"; 

    Next 

    Print Spc(4); 

    For m = 1 To 2 * n – 1 

      Print Spc(1); "*"; 

    Next 

    Print 

  Next 

End Sub 

6.30  马克思曾经做过这样一道趣味数学题：有 30 个人在一家小饭馆里用餐，其中有

男人、女人和小孩。每个男人花了 3 先令，每个女人花了 2 先令，每个小孩花了 1 先令，一

共花去 50 先令。问男人、女人和小孩各有几人？ 

【解答】设有 x 个男人，y 个女人，z 个小孩。依题意，列出以下方程组： 
30

3 2 50

x y z
x y z
  

   
 

由于两个方程式中有 3 个未知数，属于不定方程，无法直接求解。可以用“穷举法”来

进行“试根”，即将各种可能的 x, y, z 组合一一进行测试，将符合条件者输出即可。 

注意，最多只能有 16 个男人，最多只能有 24 个女人。 

窗体界面的设计如图 6-10 所示。这里给出命令按钮的

Click 事件代码： 

Private Sub Command1_Click( ) 

  List1.Clear 

  For x = 1 To 16 

    For y = 1 To 24 

      z = 30 - x - y 

      If 3 * x + 2 * y + z = 50 Then 

        p = Format(x, "@@@") & Format(y, "@@@@@") & Format(z, "@@@@@") 

        List1.AddItem p 

      End If 

    Next 

  Next 

End Sub 

6.31  利用循环语句在窗体上显示不同字号大小。 

【解答】窗体界面的设计如图 6-11 所示。Command1

的 Click 事件代码为： 

Private Sub Command1_Click( ) 

  Cls 

  Form1.FontName = "隶书" 

  For i = 10 To 15 
 

图 6-11  显示不同字号大小 

 
图 6-10  趣味数学题 

    Form1.FontSize = i 

    Print "字号大小："; i 

  Next i 

End Sub 

6.32  用 1, 2, 3, 4 这 4 个数字组成 4 位数。编写程序，打印出所有可能的 4 位数（4 个

数字可以相同），并统计出所组成的 4 位数的个数。 

【解答】这个问题可以用四重循环来解决，每层循环的初值均为 1，终值为 4，步长为 1。

总的循环次数就是所有 4 位数的个数。程序代码如下： 

Private Sub Form_Load( ) 

  Show 

  Sum = 0 

  For a = 1 To 4 

    For b = 1 To 4 

      For c = 1 To 4 

        For d = 1 To 4 

          x = a * 1000 + b * 100 + c * 10 + d 

          Print x; 

          Sum = Sum + 1 

          If Sum Mod 15 = 0 Then Print 

        Next d 

      Next c 

    Next b 

  Next a 

  Print 

  Print "和为："; Sum 

End Sub 

运行结果如图 6-12 所示。 

 
图 6-12  用 1, 2, 3, 4 组成 4 位数 

6.33  用“筛法”找出 1～100 之间的全部素数。 

【分析】“筛法”求素数表是由希腊著名数学家 Eratost henes 提出来的，其方法是：在纸

上写出 1～n 的全部整数，如图 6-13 所示（n=100）。 

然后逐一判断它们是否素数，找出一个非素数就把它挖

掉（筛掉），最后剩下的就是素数。具体做法如下。 

1）先将 1 挖掉。 

2）用 2 为除数去除它后面的每个数，即把能被 2 整除

的数挖掉，即把 2 的倍数挖掉。 

3）用 3 为除数去除它后面的每个数，即把 3 的倍数挖

掉。 

4）分别用 4, 5…各数作为除数去除这些数后面的各数（4 已被挖掉，不必再用 4 作为除

数，需用未被挖掉的数作为除数）。这个过程一直进行到除数为 n 为止（如果 n 不是整数

就取其整数部分），如图 6-14 所示。 

    
（a）                                        （b） 

              
（c）                                             （d） 

图 6-14  用“筛法”找素数 

5）经过如此“筛法”，剩下的部分全部是素数。 

【解答】设计步骤如下。 

1）建立应用程序用户界面。新建一个工程，进入窗体设计器，首先增加两个命令按钮

Command1, Command2，一个图片控件 Picture1 作为容器，然后选中 Picture1，在其中增加一

个标签控件 Label1，如图 6-15 所示。 

2）设置对象属性。 

修改窗体的 Caption 属性为：用“筛法”找素数。 

修改 Command1 的 Caption 属性为：开始。 

修改 Command2 的 Caption 属性为：重置。 

修改 Picture1 的 BackColor 属性为：（白色）。 

修改 Label1 的 BackColor 属性为：（白色），Height 属性为：260，Width 属性为：280。 

 
图 6-13  界面和运行结果 


